


Part 1: Practical Basic Verilog Examples

Erwin


Verilog Module

- **Combinational:** output depends only on the current inputs — no memory, no clock
- **Sequetial:** output depends on current inputs and past states. It requires a clock to update its state.



Combinational Circuit Examples


Multiplexer using Assign



mux_2to1.v

```
1  module mux_2to1
2      #(
3          parameter WIDTH = 8
4      )
5      (
6          input wire [WIDTH-1:0] a,
7          input wire [WIDTH-1:0] b,
8          input wire [0:0] sel,
9          output wire [WIDTH-1:0] y
10     );
11
12     assign y = (sel == 1'b0) ? a : b;
13
14 endmodule
```


Multiplexer using Always




```
1  module mux_2to1
2      #(
3          parameter WIDTH = 8
4      )
5      (
6          input wire [WIDTH-1:0] a,
7          input wire [WIDTH-1:0] b,
8          input wire [0:0] sel,
9          output reg [WIDTH-1:0] y
10     );
11
12     always @(*)
13         if (sel == 1'b0)
14             y = a;
15         else
16             y = b;
17
18 endmodule
```

Multiplexer Testbench

```
mux_2to1_tb.v
1  `timescale 1ns / 1ps
2
3  module mux_2to1_tb();
4      localparam T = 10;
5
6      reg [7:0] a, b;
7      reg [0:0] sel;
8      wire [7:0] y;
9
10     mux_2to1#(.WIDTH(8))
11     dut(.a(a), .b(b), .sel(sel), .y(y));
12
13     initial
14     begin
15         a = 8; b = 16;
16
17         sel = 0; #T;
18         sel = 1; #T;
19
20         a = 0; b = 0;
21     end
22 endmodule
```




Shifter



shifter.v

```
1  module shifter
2      (
3          input wire [7:0] a,
4          input wire      dir, // 0: left, 1: right
5          input wire [1:0] amt,
6          output wire [7:0] y
7      );
8
9      reg [7:0] y_tmp;
10
11     // Module body using "always block"
12     always @(a or dir or amt) // or
13     // always @(*) // Wildcard, produce the same result
14     begin
15         case ({dir, amt})
16             3'b000: y_tmp = {a[6:0], 1'b0};
17             3'b001: y_tmp = {a[5:0], 2'b00};
18             3'b010: y_tmp = {a[4:0], {3{1'b0}}}; // Replicate bit
19             3'b011: y_tmp = {a[3:0], 4'b0000};
20             3'b100: y_tmp = {1'b0, a[7:1]};
21             3'b101: y_tmp = {2'b00, a[7:2]};
22             3'b110: y_tmp = {3'b000, a[7:3]};
23             3'b111: y_tmp = {4'b0000, a[7:4]};
24             default: y_tmp = 8'h00;
25         endcase
26     end
27
28     // Module body using continuous assignment
29     assign y = y_tmp;
30
31 endmodule
```

Shifter Testbench



shifter_tb.v

```

1  `timescale 1ns / 1ps
2
3  module shifter_tb();
4      localparam T = 10;
5
6      reg [7:0] a;
7      reg dir;
8      reg [1:0] amt;
9      wire [7:0] y;
10
11     shifter dut(.a(a), .dir(dir), .amt(amt), .y(y));
12
13     initial
14     begin
15         dir = 0;
16         a = 8'b10101100; amt = 0; #T;
17         a = 8'b10101100; amt = 1; #T;
18         a = 8'b10101100; amt = 2; #T;
19         a = 8'b10101100; amt = 3; #T;
20
21         dir = 1;
22         a = 8'b10101100; amt = 0; #T;
23         a = 8'b10101100; amt = 1; #T;
24         a = 8'b10101100; amt = 2; #T;
25         a = 8'b10101100; amt = 3; #T;
26
27         a = 0;
28         amt = 0;
29     end
30 endmodule


```

> a[7:0]	10101100	10101100							
dir	0								
> amt[1:0]	0	0	1	2	3	0	1	2	3
> y[7:0]	01011000	01011000	10110000	01100000	11000000	01010110	00101011	00010101	00001010


Math Look-Up Table

$$\sigma(x) = \frac{1}{1 + e^{-x}}$$

x	$\sigma(x)$
...	...
-0.1250	0.4687500000
-0.0625	0.4843750000
+0.0000	0.5000000000
+0.0625	0.5146484375
...	...




Math Look-Up Table



lut_sigmoid.v


```
1  module lut_sigmoid
2      (
3          input wire          en,
4          input wire [7:0]    x, // Fixed-point: 1 sign, 3 integer, 4 fraction
5          output wire [15:0] sig // Fixed-point: 1 sign, 5 integer, 10 fraction
6      );
7
8      reg [15:0] sig_mem;
9
10     // Module body using "always block"
11     always @(x)
12     begin
13         case (x)
14             // Add more ...
15             8'b1_111_1110: sig_mem = 16'b0_00000_0111100000; // sig(-0.1250)
16             8'b1_111_1111: sig_mem = 16'b0_00000_0111110000; // sig(-0.0625)
17             8'b0_000_0000: sig_mem = 16'b0_00000_1000000000; // sig(+0.0000)
18             8'b0_000_0001: sig_mem = 16'b0_00000_1000001111; // sig(+0.0625)
19             // Add more ...
20             default: sig_mem <= 16'h0000;
21         endcase
22     end
23
24     // Module body using continuous assignment
25     assign sig = (en == 1'b1) ? sig_mem : 16'h0000;
26     // Module body using module instantiation (produce the same result)
27     // mux_2to1 #(16) mux_2to1_0(16'h0000, sig_mem, en, sig);
28
29 endmodule
```

Multiply and Add



```
pe.v
1  module pe
2      #(
3          parameter WIDTH = 16,
4          parameter FRAC_BIT = 10
5      )
6      (
7          input wire signed [WIDTH-1:0] a_in,
8          input wire signed [WIDTH-1:0] y_in,
9          input wire signed [WIDTH-1:0] b,
10         output wire signed [WIDTH-1:0] a_out,
11         output wire signed [WIDTH-1:0] y_out
12     );
13
14     wire signed [WIDTH*2-1:0] y_out_i;
15
16     assign a_out = a_in;
17     assign y_out_i = a_in * b;
18     assign y_out = y_in + y_out_i[WIDTH+FRAC_BIT-1:FRAC_BIT];
19
20 endmodule
```

Multiply and Add



pe_tb.v

```

1  `timescale 1ns / 1ps
2
3  module pe_tb();
4      localparam T = 10;
5
6      reg signed [WIDTH-1:0] a_in, y_in, b;
7      wire signed [WIDTH-1:0] a_out, y_out;
8
9      pe#(.WIDTH(16), .FRAC_BIT(10))
10     dut(.a_in(a_in), .y_in(y_in), .b(b), .a_out(a_out), .y_out(y_out));
11
12     initial
13     begin
14         a_in = 0; y_in = 0; b = 0;
15         #T;
16         a_in = 16'b0_00000_10000000000;
17         y_in = 16'b0_00000_10000000000;
18         b = 16'b0_00001_00000000000;
19         #T;
20         a_in = 16'b0_00000_1010100001;
21         y_in = 16'b1_11110_1110000110;
22         b = 16'b0_00101_1010111100;
23         #T;
24         a_in = 0; y_in = 0; b = 0;
25         #T;
26     end
27 endmodule
    
```

> a_in[15:0]	0.0	0.0	0.5	0.6572265625	0.0
> b[15:0]	0.0	0.0	1.0	5.68359375	0.0
> y_in[15:0]	0.0	0.0	0.5	-1.119140625	0.0
> a_out[15:0]	0.0	0.0	0.5	0.6572265625	0.0
> y_out[15:0]	0.0	0.0	1.0	2.6162109375	0.0

Be Careful of Bit Growth!


- Result of Addition: bit +1
- Result of Multiplication: bit x2

Be Careful of Bit Growth!

🧩 1. Addition Example — +1 Bit Growth

When adding two `N-bit` numbers,
the result **must have N+1 bits** to hold possible carry-out.

verilog

 Copy code

```
module add_example;
    reg  [3:0] a, b;      // 4-bit inputs
    wire [4:0] sum;       // 5-bit output (4+1)

    assign sum = a + b;   // bit growth +1


    initial begin
        a = 4'b1111; // 15
        b = 4'b1111; // 15
        #1 $display("a=%d, b=%d, sum=%d (binary %b)", a, b, sum, sum);
    end
endmodule
```

Be Careful of Bit Growth!

🌱 2. Multiplication Example — Bit Width Doubles

When multiplying two **N-bit** numbers,
the result **requires 2N bits** to represent the full product.

verilog

 Copy code


```
module mult_example;
    reg  [3:0] a, b;      // 4-bit inputs
    wire [7:0] product;   // 8-bit output (4+4)

    assign product = a * b; // bit growth x2

    initial begin
        a = 4'b1111; // 15
        b = 4'b1111; // 15
        #1 $display("a=%d, b=%d, product=%d (binary %b)", a, b, product, product);
    end
endmodule
```

Sequential Circuit Examples

Register



```
module dff
(
    input wire clk,
    input wire rst_n,
    input wire d,
    output reg q
);


always @(posedge clk)
    if (!rst_n)
        q <= 0;
    else
        q <= d;

endmodule
```

clk	rst_n	q*
0	-	q
1	-	q
↑	0	0
↑	1	d

- reg is not the same as register.
- reg data type, not an explicit flip-flop or hardware register.


Register More Bits



register.v


```
1  module register
2      #(
3          parameter WIDTH = 16
4      )
5      (
6          input wire          clk,
7          input wire          rst_n,
8          input wire          en,
9          input wire          clr,
10         input wire signed [WIDTH-1:0] d,
11         output reg signed [WIDTH-1:0] q
12     );
13
14     always @(posedge clk)
15     begin
16         if (!rst_n || clr)
17         begin
18             q <= 0;
19         end
20         else if (en)
21         begin
22             q <= d;
23         end
24     end
25
26 endmodule
```

Register Testbench




register_tb.v

```
1  `timescale 1ns / 1ps
2
3  module register_tb();
4      localparam T = 10;
5
6      reg clk, rst_n, en, clr;
7      reg [7:0] d;
8      wire [7:0] q;
9
10     register #(8)
11     dut(.clk(clk), .rst_n(rst_n), .en(en), .clr(clr), .d(d), .q(q));
12
13     always
14     begin
15         clk = 0;
16         #(T/2);
17         clk = 1;
18         #(T/2);
19     end
20
21     initial
22     begin
23         en = 1; clr = 0; d = 0;
24
25         rst_n = 0; #T;
26         rst_n = 1; #T;
27
28         d = 8'd8; #T;
29         d = 8'd16; #T;
30
31         clr = 1; #T;
32     end
33 endmodule
```



Counter




counter.v

```

1  module counter
2      (
3          input wire      clk,
4          input wire      rst_n,
5          input wire      clr,
6          input wire      start,
7          output wire [3:0] q
8      );
9
10     reg [3:0] cnt_reg;
11
12     always @(posedge clk)
13     begin
14         if (!rst_n || clr)
15             begin
16                 cnt_reg <= 0;
17             end
18         else if (start)
19             begin
20                 cnt_reg <= cnt_reg + 1;
21             end
22         else if (cnt_reg >= 1 && cnt_reg <= 4)
23             begin
24                 cnt_reg <= cnt_reg + 1;
25             end
26         else if (cnt_reg >= 5)
27             begin
28                 cnt_reg <= 0;
29             end
30     end
31
32     assign q = cnt_reg;
33
34 endmodule

```

Counter




counter_tb.v

```

1  `timescale 1ns / 1ps
2
3  module counter_tb();
4      localparam T = 10;
5
6      reg clk, rst_n, clr, start;
7      wire [7:0] q;
8
9      counter dut(.clk(clk), .rst_n(rst_n), .clr(clr), .start(start), .q(q));
10
11     always
12     begin
13         clk = 0;
14         #(T/2);
15         clk = 1;
16         #(T/2);
17     end
18
19     initial
20     begin
21         clr = 0;
22         start = 0;
23
24         rst_n = 0; #T;
25         rst_n = 1; #T;
26
27         start = 1; #T;
28         start = 0;
29     end
30 endmodule

```



Common Mistakes

Variable Assigned in Multiple Always Blocks

```
reg y, a, b, clear;
```

```
always @(*)  
    if (clear)  
        y = 1'b0;
```

Wrong

```
always @(*)  
    y = a & b;
```

```
always @(*)  
    if (clear)  
        y = 1'b0;  
    else  
        y = a & b;
```

Correct

[Synth 8-3352] multi-driven net \PIO_CH[0]_RREADY with 1st driver pin 'i_7/O' [fxt_dma_packer.v:65]
[Synth 8-3352] multi-driven net \PIO_CH[0]_RREADY with 2nd driver pin 'GND' [fxt_dma_packer.v:65]
[Synth 8-3352] multi-driven net \DMA_CH[3]_WREADY with 1st driver pin 'i_8/O' [fxt_dma_packer.v:65]
[Synth 8-3352] multi-driven net \DMA_CH[3]_WREADY with 2nd driver pin 'GND' [fxt_dma_packer.v:65]
[Synth 8-3352] multi-driven net \DMA_CH[3]_WREADY with 1st driver pin 'i_8/O' [fxt_dma_packer.v:65]

Incomplete Branch and Incomplete Output Assignment

```
always @(*)
  if (a > b)          // eq is not assigned in this branch
    gt = 1'b1;
  else if (a == b) // gt is not assigned in this branch
    eq = 1'b1;
                    // final else branch is omitted
```

Wrong

```
always @(*)
  if (a > b)
    begin
      gt = 1'b1;
      eq = 1'b0;
    end
  else if (a == b)
    begin
      gt = 1'b0;
      eq = 1'b1;
    end
  else // i.e., a < b
    begin
      gt = 1'b0;
      eq = 1'b0;
    end
end
```

Correct

Incomplete Branch and Incomplete Output Assignment

```
always @(*)
  if (a > b)
    begin
      gt = 1'b1;
      eq = 1'b0;
    end
  else if (a == b)
    begin
      gt = 1'b0;
      eq = 1'b1;
    end
  else // i.e., a < b
    begin
      gt = 1'b0;
      eq = 1'b0;
    end
end
```

Correct

```
always @(*)
begin
  gt = 1'b0; // Default value for gt
  eq = 1'b0; // Default value for eq
  if (a > b)
    gt = 1'b1;
  else if (a == b)
    eq = 1'b1;
end
```

Correct (alternative)

Incomplete Branch and Incomplete Output Assignment

```
reg [1:0] s;  
  
always @(*)  
begin  
    case (s)  
        2'b00: y = 1'b1;  
        2'b10: y = 1'b0;  
        2'b11: y = 1'b1;  
    endcase  
end
```

Wrong











```
always @(*)  
begin  
    case (s)  
        2'b00: y = 1'b1;  
        2'b10: y = 1'b0;  
        2'b11: y = 1'b1;  
        default: y = 1'b0;  
    endcase  
end
```

Correct

```
always @(*)  
begin  
    y = 1'b0; // Default value for y  
    case (s)  
        2'b00: y = 1'b1;  
        2'b10: y = 1'b0;  
        2'b11: y = 1'b1;  
    endcase  
end
```


Correct

Incomplete Branch and Incomplete Output Assignment

- ▼  Synthesis (9 warnings)
 - ▼  [Synth 8-327] inferring latch for variable 'FSM_sequential_next_state_reg' [moore_det_nonov.v:18] (2 more like this)
 -  [Synth 8-327] inferring latch for variable 'FSM_onehot_next_state_reg' [moore_det_nonov.v:18]
 -  [Synth 8-327] inferring latch for variable 'FSM_onehot_next_state_reg' [moore_det_nonov.v:18]
 -  [Synth 8-7080] Parallel synthesis criteria is not met
 - ▼  [Synth 8-3332] Sequential element (SD1/FSM_onehot_next_state_reg[4]) is unused and will be removed from module seq_det. (4 more like this)
 -  [Synth 8-3332] Sequential element (SD1/FSM_onehot_next_state_reg[3]) is unused and will be removed from module seq_det.
 -  [Synth 8-3332] Sequential element (SD1/FSM_onehot_next_state_reg[2]) is unused and will be removed from module seq_det.
 -  [Synth 8-3332] Sequential element (SD1/FSM_onehot_next_state_reg[1]) is unused and will be removed from module seq_det.
 -  [Synth 8-3332] Sequential element (SD1/FSM_onehot_next_state_reg[0]) is unused and will be removed from module seq_det.


A design that has multi-driver nets and/or inferred latches may work in RTL simulation but will not work on an FPGA.


Setup and Hold Time




Timing			
Design Timing Summary			
Setup	Hold	Pulse Width	
Worst Negative Slack (WNS): -24.514 ns	Worst Hold Slack (WHS): 0.018 ns	Worst Pulse Width Slack (WPWS): 4.020 ns	
Total Negative Slack (TNS): -6205.642 ns	Total Hold Slack (THS): 0.000 ns	Total Pulse Width Negative Slack (TPWS): 0.000 ns	
Number of Failing Endpoints: 274	Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	
Total Number of Endpoints: 6286	Total Number of Endpoints: 6286	Total Number of Endpoints: 2490	
Timing constraints are not met.			

Setup and Hold Time





Desired logic function



Same logic function - pipelined version.

Conlcusion

- Combinational examples
- Sequential examples
- Common errors